
1. Overview

This document lists all the changes needed to correct problems in the TTCN implementation of EMM test case 9.2.1.2.12 which are part of the LTE test suite.

The test case can be demonstrated to run with two LTE UEs (see section 6). Execution log files are provided as evidence. 

3. Verification Test Summary

Test Cases:
TC_9_2_1_2_12
Test Group:
EMM

ATS Version:
iwd-EUTRA-B2011-03_D11wk27

System Simulator used:
Anite Conformance Toolset
UE(s) used:                       Nvidia ICERA 410, Qualcomm MDM 9600
Verification Status:
PASS
4. Corrections required for test case 9.2.1.2.12

4.1 Introduction

This section describes the changes required to make test case 9.2.1.2.12 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-B2011-03_D11wk27 release.

4.2 Change 1

Testcase name
9.2.1.2.12

Reason for change
Test case needs update as per R5-113124

Summary of change
Updated TTCN implementation as per RAN5#52 CR R5-113124

Note: As complete test case is modified, TTCN source code is provided as an attachment. 

Source of change
NAS_Attach_EOnly.ttcn

Before change:

  function f_TC_9_2_1_2_12_EUTRA() runs on EUTRA_PTC

  {

    // @desc Combined attach / Rejected / EPS services not allowed in this PLMN

    /* Test Purpose (TP)

  (1)

  with { UE has sent an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message }

    ensure that {

      when { UE receives an ATTACH REJECT message with the reject cause set to "EPS services not allowed in this PLMN" }

        then { UE stores the PLMN identity in the "forbidden PLMNs for GPRS service" list and deletes any GUTI, last visited registered TAI, KSI and enters EMM-DEREGISTERED.PLMN-SEARCH state}

  }

  (2)

  with { UE in E-UTRA EMM-DEREGISTERED.PLMN-SEARCH state, and a PLMN is stored in the "forbidden PLMNs for GPRS service" list }

    ensure that {

      when { UE detects a cell which belongs to a PLMN which is not in the "forbidden PLMNs for GPRS service" list }

        then { UE performs an attach procedure }

  }

  (3)

  with { UE is switched off when a PLMN is stored in the "forbidden PLMNs for GPRS service" list }

    ensure that {

      when { UE is powered on in a cell which belongs to this PLMN }

        then { UE performs an attach procedure }

  }

  System Simulator:

   -   Cell I, Cell J and Cell A

   -   Cell I, Cell J and Cell A belong to different PLMNs and different frequencies

  UE:

   -   The UE is configured to initiate combined EPS/IMSI attach.

   -   The test USIM contains IMSI1, GUTI1 and TAI1, and EPS update status is "EU1: UPDATED".

   -   If pc_GERAN or pc_UTRAN, the test USIM contains TMSI-1 and a valid LAI-1 for CS service, and Location update status is " U1: UPDATED".

  Preamble:

   -   The UE is in state Switched OFF (state 1) according to clause TS 36.508.

 */

    f_EUTRA_NAS_Init(LTE_CellsOnDifferentPLMN, c1);

    // create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_EUTRA_CellConfig_Def(eutra_CellI);

    f_EUTRA_CellConfig_Def(eutra_CellJ);

    f_UT_ConfigureCombinedAttach (UT); // @sic R5-103815 sic@

    f_EUTRA_Preamble(eutra_CellA, STATE2_IDLEUPDATE,Combined); // @sic R5s110176 sic@

    f_EUTRA_UE_Detach_SwitchOffUe ( eutra_CellA, RRC_IDLE,-,Combined); // @sic R5s110176 sic@

    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_1_2_12_Body ();

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellI, E2_CONNECTED, Combined);   // @sic R5s110176, R5-103875 sic@
  } // end f_TC_9_2_1_2_12_EUTRA

  function fl_TC_9_2_1_2_12_Body() runs on EUTRA_PTC

  {

    var template (value) CellPowerList_Type v_CellPowerList;

    var NAS_MSG_Indication_Type v_NAS_Ind;

    var SRB_COMMON_IND v_ReceivedAsp;

    var boolean v_EIT_Flag := false;

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    var template (value) AccessPointName v_APN;

    var template (omit) ProtocolConfigOptions v_Pco := omit;

    var HalfOctet_Type v_EpsBearerId := tsc_EpsDefaultBearerId;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var float v_NoReactionTime := f_EUTRA_SetTimerToleranceMax( eutra_CellA, rrcTimer, 3.0 );

    var GutiParameters_Type v_Guti_Params := f_EUTRA_CellInfo_GetGuti (eutra_CellA);

    var GutiParameters_Type v_Guti_Params_J := f_EUTRA_CellInfo_GetGuti (eutra_CellJ);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId (v_Guti_Params.PLMN_Identity);

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC ( eutra_CellJ );

    var template (value) TrackingAreaIdList v_TaiList := cds_TAIListNonConsecutive_tlv ( v_PLMN, { bit2oct(v_Tac) } );

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (Combined); // @sic R5s110176 sic@

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(Combined); // @sic R5s110176 sic@

    //@siclog "Step 1" siclog@

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_ServingCellRS_EPRE ),

      cs_CellPower ( eutra_CellI, tsc_NonSuitableCellRS_EPRE ),

      cs_CellPower ( eutra_CellJ, tsc_SuitableCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    //  Switch on UE

    //@siclog "Step 2" siclog@

    f_UT_SwitchOnUE (UT, false);

    // The UE transmits an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message

    //@siclog "Step 3" siclog@

    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellA,

                                                tsc_RRC_TI_Def,

                                                cr_EstablishmentCause_Any,

                                                cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                  cdr_ATTACH_REQUEST_AttachType (v_AttachType,

                                                                                                                 v_AdditionalUpdateType,

                                                                                                                 f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti_Params),

                                                                                                                 cr_TAI(v_PLMN, tsc_TAC1),

                                                                                                                 f_EUTRA_GetCellLAI(eutra_CellA, Combined), // @sic R5s110176 sic@

                                                                                                                 omit),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                                  f_GetPdnType())

                                                                                  ));

    //@siclog "Step 4" siclog@

    // The SS transmits an ATTACH REJECT message with EMM cause = "EPS services not allowed in this PLMN"

    SRB.send  ( cas_SRB1_NasPdu_REQ ( eutra_CellA,

                                      cs_TimingInfo_Now,

                                      cs_NAS_Request ( tsc_SHT_NoSecurityProtection,

                                                       cs_508_AttachReject (cs_EMM_Cause_v(tsc_Cause_EPSServNotAllowedInThisPLMN),

                                                                            omit)) ));

    // The SS releases the RRC connection

    //@siclog "Step 5" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

    //@siclog "Step 6" siclog@

    // Check: Does the test result of generic procedure in TS 36.508 subclause 6.4.2.5 indicate that the UE

    //        does not respond to paging when paged with GUTI-1 and with CN domain indicator set to "PS"?

    f_EUTRA_508Check_NASNoResponse ( eutra_CellA,

                                     cs_NasPaging_OneRecord ( cs_PagingUE_Identity_S_TMSI ( v_Guti_Params.MME_Code, v_Guti_Params.M_TMSI )),

                                     v_NoReactionTime );

    //parallel behaviour -> FFS

    // v_Px_CS_PSmode2 not read until a solution is agreed for this feature

    // since v_Px_CS_PSmode2 doesn't exist in the prose specification

    // For the time being pc_Combined_Attach is not read and the parallel behaviour is done every time

    //if(v_Px_CS_PSmode2)

    //{

    // Check: Does the UE transmit an ATTACH REQUEST message?

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellJ,

                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,

                                                                      cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                      cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                      f_GetPdnType()))))

      -> value v_ReceivedAsp;

    v_PdnConnectivityRequest := v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // Use this value in SS responses

    if (ispresent( v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions)) {

      v_Pco := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;

    }

    // The SS transmits an AUTHENTICATION REQUEST message to initiate the EPS authentication and AKA procedure

    // The UE transmits an AUTHENTICATION RESPONSE message

    // The SS transmits a NAS SECURITY MODE COMMAND message

    // The UE transmits a NAS SECURITY MODE COMPLETE message

    v_APN := f_EUTRA_Authentication_NAS_AS_Security (eutra_CellJ, v_EIT_Flag, v_EPS_TI);

    // SS responds with ATTACH ACCEPT message

    v_TaiList := cs_TAIListNonConsecutive_lv(f_Asn2Nas_PlmnId(v_Guti_Params_J.PLMN_Identity),

                                             { bit2oct(f_EUTRA_CellInfo_GetTAC(eutra_CellJ)) } );

    f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(eutra_CellJ,

                                                                  v_EpsBearerId,

                                                                  tsc_RRC_TI_Def,

                                                                  v_EPS_TI,

                                                                  v_AttachType,

                                                                  f_EUTRA_GetPDNAddress(f_CheckPCOforIPallocationViaNas(v_Pco)),

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti_Params_J),

                                                                  v_TaiList,

                                                                  omit,

                                                                  omit,

                                                                  omit,

                                                                  v_APN,

                                                                  f_GetDefPCOforPCSCFAddress (v_Pco), // @sic R5-104703 sic@

                                                                  f_GetAdditionalUpdateResult(v_PdnConnectivityRequest.aTTACH_REQUEST.addUpdateType, // @sic R5s110176 sic@

                                                                                              Combined)); // @sic R5s110176 sic@

    // Check: The UE transmits an ATTACH COMPLETE message

    f_EUTRA_NAS_AttachComplete(eutra_CellJ, v_EpsBearerId);

    // }

        //@siclog "Step 8" siclog@

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_ServingCellRS_EPRE ),

      cs_CellPower ( eutra_CellI, tsc_SuitableCellRS_EPRE ),

      cs_CellPower ( eutra_CellJ, tsc_NonSuitableCellRS_EPRE )

    }

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    //@siclog "Step 9" siclog@

    // Check: Does the UE transmit an ATTACH REQUEST message?

    //@sic R5-112625 sic@

    SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellI, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)));

    f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellI);

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellI,

                                             cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                             cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                               cdr_ATTACH_REQUEST_WithOldMobileIdNoTAI(v_AttachType, v_AdditionalUpdateType, f_Imsi2MobileIdentity(px_IMSI_Def) )

                                                               )));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9: Attach request");

    //@siclog "Step 10" siclog@

    // The SS transmits an ATTACH REJECT message with EMM cause = "EPS services not allowed in this PLMN"

    SRB.send  ( cas_SRB1_NasPdu_REQ ( eutra_CellI,

                                      cs_TimingInfo_Now,

                                      cs_NAS_Request ( tsc_SHT_NoSecurityProtection,

                                                       cs_508_AttachReject (cs_EMM_Cause_v(tsc_Cause_EPSServNotAllowedInThisPLMN),

                                                                            omit))

                                      ));

    //  The SS releases the RRC connection

    //@siclog "Step 11" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellI);

    //  The UE is switched off

    //@siclog "Step 12" siclog@

    f_UT_SwitchOffUE(UT, false, CNF_REQUIRED); // @sic R5s1100395 sic@

    v_CellPowerList := {

      cs_CellPower ( eutra_CellA, tsc_NonSuitableOffCellRS_EPRE ),

      cs_CellPower ( eutra_CellI, tsc_ServingCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    //@siclog "Step 13" siclog@

    // The UE is switched on.

    f_UT_SwitchOnUE (UT, false);

    //@siclog "Step 14" siclog@

    // Check: Does the UE transmit an ATTACH REQUEST message on Cell I?

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellI,

                                    cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,

                                                                      cdr_ATTACH_REQUEST_WithOldMobileIdNoTAI(v_AttachType, v_AdditionalUpdateType, f_Imsi2MobileIdentity(px_IMSI_Def) ),

                                                                      cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                      f_GetPdnType())

                                                                      )))

      -> value v_ReceivedAsp;

    v_PdnConnectivityRequest := v_ReceivedAsp.Signalling.Nas[0].Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier; // Use this value in SS responses

    // Get the PCO to be used later (TS 36.508 v.8.2.2 Table 4.7.3-6)

    if (ispresent( v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions)) {

      v_Pco := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;

    }

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 14");

    v_EIT_Flag := f_CheckEsmInfoTransferFlag(v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST);

    //@siclog "Step 15 - 26" siclog@

    // The SS transmits an AUTHENTICATION REQUEST message to initiate the EPS authentication and AKA procedure.

    // The UE transmits an AUTHENTICATION RESPONSE message and establishes mutual authentication.

    // The SS transmits a NAS SECURITY MODE COMMAND message to activate NAS security.

    // The UE transmits a NAS SECURITY MODE COMPLETE message and establishes the initial security configuration.

    v_APN := f_EUTRA_Authentication_NAS_AS_Security (eutra_CellI, v_EIT_Flag, v_EPS_TI);

    // SS responds with ATTACH ACCEPT message with IE EPS Bearer Identity set to default EPS Bearer Context.

    // The ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST message is piggybacked in ATTACH ACCEPT message

    v_TaiList := cs_TAIListNonConsecutive_lv(f_Asn2Nas_PlmnId(v_Guti_Params.PLMN_Identity),

                                             { bit2oct(f_EUTRA_CellInfo_GetTAC(eutra_CellI)) } );

    f_EUTRA_RRCConnectionReconfigurationWithPiggyNAS(eutra_CellI,

                                                     tsc_RRC_TI_Def,

                                                     cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                                                    cs_508_ATTACH_ACCEPT(tsc_EAttachType_EpsOnly,

                                                                                                         v_TaiList,

                                                                                                         omit,

                                                                                                         omit,

                                                                                                         omit,

                                                                                                         cs_EMM_Cause_tv(tsc_Cause_CSDomainNotAvailable),

                                                                                                         f_GetAdditionalUpdateResult(v_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.addUpdateType,Combined)),  // @sic R5s110176, R5-103681 sic@

                                                                                    cs_508_ActivateDefEpsBearerContextRequest(v_EpsBearerId,

                                                                                                                              v_EPS_TI,

                                                                                                                              v_APN,

                                                                                                                              f_EUTRA_GetPDNAddress(f_CheckPCOforIPallocationViaNas(v_Pco)),

                                                                                                                              omit,

                                                                                                                              omit,

                                                                                                                              f_GetDefPCOforPCSCFAddress (v_Pco))));

    // Check: Does the UE transmit an ATTACH COMPLETE message including an ACTIVATE DEFAULT EPS BEARER

    // CONTEXT ACCEPT message?

    f_EUTRA_NAS_AttachComplete(eutra_CellI, v_EpsBearerId);
  }// end function fl_TC_9_2_1_2_12_Body

After change:

function f_TC_9_2_1_2_12_EUTRA() runs on EUTRA_PTC

  {

    // @desc Combined attach / Rejected / EPS services not allowed in this PLMN

    /* Test Purpose (TP)

  (1)

  with { UE has sent an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message }

  ensure that {

    when { UE receives an ATTACH REJECT message with the reject cause set to "EPS services not allowed in this PLMN" }

      then { UE deletes any GUTI, last visited registered TAI, KSI and enters EMM-DEREGISTERED.PLMN-SEARCH state}

              }

  (2)

  with { UE in E-UTRA EMM-DEREGISTERED.PLMN-SEARCH state, and a PLMN is stored in the "forbidden PLMNs for GPRS service list" }

  ensure that {

    when { UE detects a cell which belongs to a PLMN which is in the "forbidden PLMNs for GPRS service list" }

      then { UE doesn’t perform an attach procedure }

              }

  (3)

  with { UE in E-UTRA EMM-DEREGISTERED.PLMN-SEARCH state, and a PLMN is stored in the "forbidden PLMNs for GPRS service list" }

  ensure that {

    when { UE detects a cell which belongs to a PLMN which is not in the "forbidden PLMNs for GPRS service list" }

      then { UE performs an attach procedure }

              }

  (4)

  with { UE is switched off when a PLMN is stored in the "forbidden PLMNs for GPRS service list" }

  ensure that {

    when { UE is powered on a cell which belongs to this PLMN }

      then { UE performs an attach procedure }

              }

  (5)

  with { UE in E-UTRA EMM-DEREGISTERED.PLMN-SEARCH state, and a PLMN is stored in the "forbidden PLMNs for GPRS service list" }

  ensure that {

    when { UE is in the cell which belongs to the rejected PLMN and when that PLMN is selected manually }

      then { UE performs an attach procedure }

  System Simulator:

  -
cell G, Cell H and cell I are configured according to Table 6.3.2.2-1 in; [18].

  -
cell G and Cell H with MCC-1/MNC-2 (visited PLMN, different TAs)

  -
Cell I with MCC-2/MNC-101 (visited PLMN)

  -
the cells may not be simultaneously activated.

  UE:

  - 
the UE is configured to initiate Combined attach;

  - 
the UE is previously registered on E-UTRAN, and when on E-UTRAN, the UE is last authenticated and registered on cell G using default message contents according to TS 36.508 [18];

  -
the "forbidden PLMNs for GPRS service list" is empty.

  -
the different cells may not be simultaneously activated (at most 2 cells are active simultaneously).

  Preamble:

  -
the UE is in state Switched OFF (state 1) according to TS 36.508 [18].

 */

    f_EUTRA_NAS_Init(LTE_CellsOnDifferentPLMN, c1);

    // create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellG);

    f_EUTRA_CellConfig_SRBs_Only(eutra_CellH);

    f_EUTRA_CellConfig_Def(eutra_CellI);

    f_UT_ConfigureCombinedAttach (UT); // @sic R5-103815 sic@

    f_EUTRA_Preamble(eutra_CellG, STATE2_IDLEUPDATE,Combined); // @sic R5s110176 sic@

    f_EUTRA_UE_Detach_SwitchOffUe ( eutra_CellG, RRC_IDLE,-,Combined); // @sic R5s110176 sic@

    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_1_2_12_Body ();

    f_EUTRA_TestBody_Set(false);

    f_UT_AutomaticPLMN_Select ( UT );

    //wait 300 seconds to let UE complete automatic PLMN search

    f_Delay(300.0);

    f_EUTRA_Postamble(eutra_CellG, E1_IDLE, Combined);   // @sic R5s110176, R5-103875 sic@
  } // end f_TC_9_2_1_2_12_EUTRA

  function fl_TC_9_2_1_2_12_Body() runs on EUTRA_PTC

  {

    var template (value) CellPowerList_Type v_CellPowerList;

    var NAS_MSG_Indication_Type v_NAS_Ind;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (Combined); // @sic R5s110176 sic@

    var GutiParameters_Type v_GutiParams_CellG := f_EUTRA_CellInfo_GetGuti(eutra_CellG);

    var NAS_AttDetValue_Type v_AttachType := f_GetEAttachType(Combined); // @sic R5s110176 sic@

    //timer t_Wait := f_EUTRA_SetTimerToleranceMax( eutra_CellA, nonProtocolTimer, 300.0 ); // This needs a prose CR @sic R5s100784 sic@!!!

    //@siclog "Step 1" siclog@

    v_CellPowerList := {

      cs_CellPower ( eutra_CellG, tsc_ServingCellRS_EPRE ),

      cs_CellPower ( eutra_CellH, tsc_NonSuitableCellRS_EPRE ),

      cs_CellPower ( eutra_CellI, tsc_NonSuitableOffCellRS_EPRE )  // @sic R5-110741 sic@

    }

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    //@siclog "Step 2 Void" siclog@

    // Switch on UE

    //@siclog "Step 3" siclog@

    f_UT_SwitchOnUE (UT, false);

    //  The UE transmits an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message

    //@siclog "Step 4" siclog@   

    v_NAS_Ind:= f_EUTRA_RRC_ConnEst_DefWithNas(eutra_CellG,

                                               tsc_RRC_TI_Def,

                                               cr_EstablishmentCause_Any,

                                               cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                 cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                                 cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                                 f_GetPdnType())));

    //@siclog "Step 5" siclog@

    // The SS transmits an ATTACH REJECT message with EMM cause = "EPS services not allowed in this PLMN"

    SRB.send  ( cas_SRB1_NasPdu_REQ ( eutra_CellG,

                                      cs_TimingInfo_Now,

                                      cs_NAS_Request ( tsc_SHT_NoSecurityProtection,

                                                       cs_508_AttachReject (

                                                                            cs_EMM_Cause_v(tsc_Cause_EPSServNotAllowedInThisPLMN),

                                                                            omit))));

    //  The SS releases the RRC connection

    //@siclog "Step 6" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellG);

    //@siclog "Step 7" siclog@

    // Check: Does the UE transmit an ATTACH REQUEST message in the next 90 seconds?

    f_EUTRA_CheckNoAttachOn1Cell ( eutra_CellG, 90.0, "Step 7" );

    //@siclog "Step 8" siclog@

    // Note: cellG and cellH are in the different TAI and same PLMN

    v_CellPowerList := {

      cs_CellPower ( eutra_CellG, tsc_NonSuitableCellRS_EPRE ),

      cs_CellPower ( eutra_CellH, tsc_ServingCellRS_EPRE )

    }

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    //@siclog "Step 9 Void" siclog@

    // Check: Does the UE transmit an ATTACH REQUEST message in the next 90 seconds?

    f_EUTRA_CheckNoAttachOn1Cell ( eutra_CellH, 90.0, "Step 9" );

    //@siclog "Step 10" siclog@

    // Note: cellG and cellI are different PLMNs

    v_CellPowerList := {

      cs_CellPower ( eutra_CellI, tsc_ServingCellRS_EPRE ),

      cs_CellPower ( eutra_CellH, tsc_NonSuitableOffCellRS_EPRE )

    }

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    //@siclog "Step 11 Void" siclog@

    //  Does the UE transmit an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message

    v_NAS_Ind:= f_EUTRA_RRC_ConnEst_DefWithNas( eutra_CellI,

                                                tsc_RRC_TI_Def,

                                                cr_EstablishmentCause_Any,

                                                cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,

                                                                                  cdr_ATTACH_REQUEST_IMSIonly(v_AttachType, v_AdditionalUpdateType),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                                  f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11");

    //@siclog "Step 12-24" siclog@

    // The SS completes the attach procedure successfully and then releases the RRC

    // connection by executing steps 5 to 17 of UE registration procedure in TS 36.508 clause 4.5.2.3

    f_EUTRA_IdleUpdated_Step5_17 (eutra_CellI, STATE2_IDLEUPDATE, Combined, v_NAS_Ind);

    // If possible (see ICS) switch off is performed. Otherwise the power is removed.

    //@siclog "Step 25" siclog@

    f_EUTRA_UE_Detach_SwitchOffUe (eutra_CellI, RRC_IDLE,-, Combined);   // @sic R5s110176 sic@ This needs a prose CR R5s100784 !!!

    //@siclog "Step 26" siclog@

    // The SS reconfigures:

    // Cell G as the "Serving cell". Cell I as a " Non Suitable cell",

    v_CellPowerList := {

      cs_CellPower ( eutra_CellG, tsc_ServingCellRS_EPRE ),

      cs_CellPower ( eutra_CellI, tsc_NonSuitableCellRS_EPRE )

    };

    f_EUTRA_SetCellPowerList ( v_CellPowerList );

    //@siclog "Step 27" siclog@

    // The UE is brought back to operation.

    f_UT_SwitchOnUE (UT, false );

    //@siclog "Step 28" siclog@

    // Check: Does the UE transmit an ATTACH REQUEST message?

    // @sic R5s110168 change 2.13 sic@

    SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellG, cr_RRCConnectionRequest(cr_EstablishmentCause_Any)));

    f_EUTRA_RRC_ConnectionSetup_Def (eutra_CellG, cs_TimingInfo_Now, tsc_RRC_TI_Def);

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellG,

                                             cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                             cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                              cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                              cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "step 28");

                                                                                                              f_GetPdnType()))));

    //@siclog "Step 29" siclog@

    // The SS transmits an ATTACH REJECT message with EMM cause = "EPS services not allowed in this PLMN"

    SRB.send  ( cas_SRB1_NasPdu_REQ ( eutra_CellG,

                                      cs_TimingInfo_Now,

                                      cs_NAS_Request ( tsc_SHT_NoSecurityProtection,

                                                       cs_508_AttachReject (cs_EMM_Cause_v(tsc_Cause_EPSServNotAllowedInThisPLMN),

                                                                            omit))));

    //  The SS releases the RRC connection

    //@siclog "Step 30" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellG);

    //  Select PLMN of cell G (as set in fl_EUTRA_InitNAS_forNAS_Cells)

    //@siclog "Step 31" siclog@

    f_UT_ManualPLMN_Select (UT, f_Asn2Nas_PlmnId(v_GutiParams_CellG.PLMN_Identity)); // @sic R5-106675 sic@

    //  Does the UE transmit an ATTACH REQUEST message including a PDN CONNECTIVITY REQUEST message

    //@siclog "Step 32" siclog@

    v_NAS_Ind:= f_EUTRA_RRC_ConnEst_DefWithNas( eutra_CellG,

                                                tsc_RRC_TI_Def,

                                                cr_EstablishmentCause_Any,

                                                cr_NAS_IndicationWithPiggybacking(tsc_SHT_NoSecurityProtection,

                                                                                  cdr_ATTACH_REQUEST_IMSIonly(v_AttachType, v_AdditionalUpdateType),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

                                                                                                                  f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 32");

    //@siclog "Steps 33- 35" siclog@

    // The attach procedure is completed and the RRC connection is released

    // by executing steps 5 to 17 of TS 36.508 sub clause 4.5.2.3.

    f_EUTRA_IdleUpdated_Step5_17 (eutra_CellG, STATE2_IDLEUPDATE, Combined, v_NAS_Ind);
  }// end function fl_TC_9_2_1_2_12_Body

  

4.3 Change 2

Testcase name
9.2.1.2.12

Reason for change
TC execution time is much more than 360s

Summary of change
Changed the Guard timer to 1200 s.

Source of change
LTE_EPS_TS_Testcases_11wk27

Before change:

  testcase TC_9_2_1_2_12() runs on MTC system SYSTEM {

    // @purpose

    //   Combined attach / Rejected / EPS services not allowed in this PLMN

    var EUTRA_PTC   v_EUTRA     := EUTRA_PTC.create alive;

    var UTRAN_PTC   v_UTRAN     := null;

    var GERAN_PTC   v_GERAN     := null;

    var CDMA2000_PTC v_CDMA2000  := null;

    timer t_GuardTimer := int2float(360);
    f_MTC_ConnectPTCs(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000);

    v_EUTRA.start   (f_TC_9_2_1_2_12_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }

After change:

  testcase TC_9_2_1_2_12() runs on MTC system SYSTEM {

    // @purpose

    //   Combined attach / Rejected / EPS services not allowed in this PLMN

    var EUTRA_PTC   v_EUTRA     := EUTRA_PTC.create alive;

    var UTRAN_PTC   v_UTRAN     := null;

    var GERAN_PTC   v_GERAN     := null;

    var CDMA2000_PTC v_CDMA2000  := null;

    timer t_GuardTimer := int2float(1200);//Anite Change
      f_MTC_ConnectPTCs(v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000);

    v_EUTRA.start   (f_TC_9_2_1_2_12_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }

5. Execution Log Files.

5.1 Nvidia Icera 410

The Nvidia Icera 410 UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_9_2_1_2_12_Nvidia.htm:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

5.2 Qualcomm MDM 9600

The Qualcomm MDM 9600 UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_9_2_1_2_12_Qualcomm.htm:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6. References

[1]
R5s110562: This archive comprises html format execution log file.

