
Overview

This document lists a TTCN change needed to correct a problem in ATS iWD-EUTRA-B2011-03_D11wk37

Corrections 

Change 1 – Correction to fl_IpInfo2MatchingInfo 

Function name
fl_IpInfo2MatchingInfo

Reason for change
Incorrect checking of v_SocketInfo allows access to “v_SocketInfo.IpAddr” when “v_SocketInfo” = omit

Summary of change
Correct checking of v_SocketInfo value.

TTCN module
IP_PCT_Routing.ttcn   

MCC160 Comment


Before change

function fl_IpInfo2MatchingInfo(template (omit) IP_Socket_Type p_SocketInfo) return template IP_Socket_Type

  {

    var template IP_Socket_Type v_SocketInfo := p_SocketInfo;

    if (isvalue(v_SocketInfo)) {
      v_SocketInfo := *;        // omit it means that the addr is "don't care"
    } else {

      if (not ispresent(v_SocketInfo.IpAddr)) {

        v_SocketInfo.IpAddr := *;

      }

      if (not ispresent(v_SocketInfo.Port)) {

        v_SocketInfo.Port := *;

      }

    }

    return v_SocketInfo;

  }



After change

function fl_IpInfo2MatchingInfo(template (omit) IP_Socket_Type p_SocketInfo) return template IP_Socket_Type

  {

    var template IP_Socket_Type v_SocketInfo := p_SocketInfo;

    if (not isvalue(v_SocketInfo)) {

      v_SocketInfo := *;        // omit it means that the addr is "don't care"
    } else {

      if (not ispresent(v_SocketInfo.IpAddr)) {

        v_SocketInfo.IpAddr := *;

      }

      if (not ispresent(v_SocketInfo.Port)) {

        v_SocketInfo.Port := *;

      }

    }

    return v_SocketInfo;

  }



Change 2 – Correction to f_Convert_OctString2IPv6Addr
Function name
f_Convert_OctString2IPv6Addr

Reason for change
Current loop allows access to out of limit indexes in the octet string type “p_IPv6AddrOct”

Summary of change
The condition to leave the “for” loop moved to the beginning of the loop.

TTCN module
CommonIP.ttcn       

MCC160 Comment


Before change

Before:

function f_Convert_OctString2IPv6Addr(O16_Type p_IPv6AddrOct) return charstring

  {

    var charstring v_IPv6Addr := "";

    var integer i;

    for (i:=0; true; i:=i+2) {

      v_IPv6Addr := v_IPv6Addr & oct2str(substr(p_IPv6AddrOct, i, 2));

      if (i < lengthof(p_IPv6AddrOct)) {
        v_IPv6Addr := v_IPv6Addr & ":";

      } else {

        break;

      }
    }

    return v_IPv6Addr;

  }



After change

function f_Convert_OctString2IPv6Addr(O16_Type p_IPv6AddrOct) return charstring

  {

    var charstring v_IPv6Addr := "";

    var integer i;

   for (i:=0; i < lengthof(p_IPv6AddrOct); i:=i+2) {

      v_IPv6Addr := v_IPv6Addr & oct2str(substr(p_IPv6AddrOct, i, 2));

      if (i < (lengthof(p_IPv6AddrOct) - 2)) {

        v_IPv6Addr := v_IPv6Addr & ":";

      }

    }

    return v_IPv6Addr;

  }



Change 3  – Correction to fl_IP_Servers_Stop
Function name
fl_IP_Servers_Stop

Reason for change
Wrong template used: “cs_IPv4Addr” instead of  “cs_IPv6Addr”

Summary of change
Template  “cs_IPv4Addr”  replaced by  “cs_IPv6Addr”

TTCN module
IP_PTC_Main.ttcn       

MCC160 Comment


Before change

function fl_IP_Servers_Stop(integer p_Index) runs on IP_PTC

  { /* @sic R5-113734 change 8 sic@ */

    var charstring v_ServerAddr;

    if (fl_IP_Server_IsRunning(vc_IP_PTC_PDN_Server.DHCP[p_Index])) {

      v_ServerAddr := vc_IP_PTC_PDN_Server.DHCP[p_Index].ServerAddr;

      fl_UdpServer_Stop(cs_IPv4Addr(v_ServerAddr), tsc_PortNumber_DHCP_Server);

      vc_IP_PTC_PDN_Server.DHCP[p_Index] := valueof(cs_IP_Server_Init);

    }

    if (fl_IP_Server_IsRunning(vc_IP_PTC_PDN_Server.ICMPv6[p_Index])) {

      v_ServerAddr := vc_IP_PTC_PDN_Server.ICMPv6[p_Index].ServerAddr;

      fl_Icmpv6Server_Stop(cs_IPv6Addr(v_ServerAddr));

      v_ServerAddr := fl_IPv6AddrGetLinkLocalAddr(v_ServerAddr);

      fl_Icmpv6Server_Stop(cs_IPv4Addr(v_ServerAddr));

      vc_IP_PTC_PDN_Server.ICMPv6[p_Index] := valueof(cs_IP_Server_Init);

    }

  }



After change

function fl_IP_Servers_Stop(integer p_Index) runs on IP_PTC

  { /* @sic R5-113734 change 8 sic@ */

    var charstring v_ServerAddr;

    if (fl_IP_Server_IsRunning(vc_IP_PTC_PDN_Server.DHCP[p_Index])) {

      v_ServerAddr := vc_IP_PTC_PDN_Server.DHCP[p_Index].ServerAddr;

      fl_UdpServer_Stop(cs_IPv4Addr(v_ServerAddr), tsc_PortNumber_DHCP_Server);

      vc_IP_PTC_PDN_Server.DHCP[p_Index] := valueof(cs_IP_Server_Init);

    }

    if (fl_IP_Server_IsRunning(vc_IP_PTC_PDN_Server.ICMPv6[p_Index])) {

      v_ServerAddr := vc_IP_PTC_PDN_Server.ICMPv6[p_Index].ServerAddr;

      fl_Icmpv6Server_Stop(cs_IPv6Addr(v_ServerAddr));

      v_ServerAddr := fl_IPv6AddrGetLinkLocalAddr(v_ServerAddr);

      fl_Icmpv6Server_Stop(cs_IPv6Addr(v_ServerAddr));

      vc_IP_PTC_PDN_Server.ICMPv6[p_Index] := valueof(cs_IP_Server_Init);

    }

  }

